## EXERCISE 3:
## Eigenvalues, eigenstates, and symmetries of the XXZ Hamiltonian

```
Clear[L, upspins, downspins, dim];
L = 6;
upspins = L / 2;
downspins = L - upspins;
dim = L! / (upspins! downspins!);

(* BASIS *)
Clear[onebasisvector, basis];
onebasisvector = Flatten[{Table[1, {k, 1, upspins}], Table[0, {k, 1, downspins}]}];
basis = Permutations[onebasisvector];


(* PARAMETERS OF THE HAMILTONIAN *)
Clear[Jxy, Jz];
Jxy = 1.0;
Jz = 0.4;


(* ELEMENTS OF THE HAMILTONIAN *)
Clear[HH];
(* Initialization *)
Do[Do[HH[i, j] = 0., {i, 1, dim}], {j, 1, dim}];

(* Diagonal elements *)
Do[
  (* Ising interaction *)
  Do[If[basis[[i, j]] == basis[[i, j + 1]],
      HH[i, i] = HH[i, i] + Jz / 4., HH[i, i] = HH[i, i] - Jz / 4.];
    , {j, 1, L - 1}];
  , {i, 1, dim}];

(* Off-diagonal elements *)
Clear[howmany, site];
Do[
  Do[
    (* Initialization *)
    howmany = 0;
    Do[site[kk] = 0, {kk, 1, L}];
    (* Sites where states i and j differ *)
    Do[ If[basis[[i, k]] ≠ basis[[j, k]], {howmany = howmany + 1, site[howmany] = k}];
     , {k, 1, L}];
    (* If only two neighbor sites differ, there is a coupling matrix element *)
    If[howmany == 2,
     If[site[2] - site[1] == 1, {HH[i, j] = Jxy / 2., HH[j, i] = Jxy / 2.}]];
    , {j, i + 1, dim}];
  , {i, 1, dim - 1}];


(* TOTAL HAMILTONIAN AND DIAGONALIZATION *)
Clear[Hamiltonian, Ene, Vec];
Hamiltonian = Table[Table[HH[i, j], {j, 1, dim}], {i, dim}];
Ene = Eigenvalues[Hamiltonian];
Vec = Eigenvectors[Hamiltonian];

(* EIGENVALUES IN INCREASING ORDER *)
Clear[Energy, Vector];
```

```
aux = Sort[Table[{Ene[[k]], k}, {k, 1, dim}]];
Energy = Sort[Ene];
Vector = Table[Vec[[aux[[k, 2]]]], {k, 1, dim}];



(* PARITY *)
(* First we find all pairs of equivalent site-basis vectors *)
(* Example: state 1100 pairs with state 0011 *)
Clear[i, j, k, pair];
Do[
  Do[Clear[mirror];
    mirror = 0;
    Do[mirror = mirror + Mod[basis[[i, k]] + basis[[j, L + 1 - k]], 2];
     , {k, 1, L}];
    If[mirror == 0, pair[i] = j];
    , {j, 1, dim}];
  , {i, 1, dim}];

(* Now we find the eigenvalues of the parity operator. We
 also count how many states are even and how many are odd *)
Clear[Neven, Nodd, parity];
(* Neven is the number of states with even parity *)
(* Nodd is the number of states with odd parity *)
Neven = 0;
Nodd = 0;
Do[
  Clear[project];
  project = 0;
  Do[ project = project + (1 / 4.) (Vector[[i]][[j]] + Vector[[i]][[pair[j]]])^2;
   , {j, 1, dim}];
  (* project will be very close to 1 if the parity is even *)
  (* project will be very close to 0 if the parity is odd *)
  If[1. - project < 0.001,
   {Neven = Neven + 1, parity[i] = +1}, {Nodd = Nodd + 1, parity[i] = -1}];
  , {i, 1, dim}];


(* ROTATION in x *)
Clear[pair];
Do[
  Do[
    Clear[rotx];
    rotx = Sum[Mod[basis[[i, k]] + basis[[j, k]], 2], {k, 1, L}];
    If[rotx == L, pair[i] = j];
    , {j, 1, dim}];
  , {i, 1, dim}];

(* Now we find the eigenvalues of the rotation operator. *)
Clear[rotationX];
Do[
  Clear[rotrot];
  rotrot = 0;
  Do[ rotrot = rotrot + (1 / 4.) (Vector[[i]][[j]] + Vector[[i]][[pair[j]]])^2;
   , {j, 1, dim}];
  (* rotrot will be very close to 1 if the rotationX is +1 *)
  (* rotrot will be very close to 0 if the rotationX is -1 *)
  If[1. - rotrot < 0.001, rotationX[i] = +1, rotationX[i] = -1];
  , {i, 1, dim}];


(* TABLE *)
tab = Table[{Energy[[i]], parity[i], rotationX[i]}, {i, 1, dim}];
TableForm[tab, TableHeadings → {None, {"Energy", "Π", "R_π^x"}}, TableAlignments → Right]
```

| Energy | $\Pi$ | $R_{\pi}^{x}$ |
|---|---|---|
| −2.02384 | −1 | −1 |
| −1.5253 | 1 | 1 |
| −1.07293 | −1 | 1 |
| −0.992344 | −1 | −1 |
| −0.75914 | 1 | 1 |
| −0.649385 | 1 | −1 |
| −0.637126 | 1 | 1 |
| −0.349111 | −1 | 1 |
| −0.216907 | −1 | −1 |
| −0.137975 | −1 | −1 |
| 0.0248642 | 1 | 1 |
| 0.0521527 | 1 | −1 |
| 0.337462 | 1 | 1 |
| 0.410489 | −1 | −1 |
| 0.60198 | −1 | −1 |
| 0.622043 | −1 | 1 |
| 0.797232 | 1 | −1 |
| 0.852698 | 1 | 1 |
| 1.1586 | −1 | −1 |
| 1.50654 | 1 | 1 |